Rules for integrands involving polylogarithms

1. $\int u\,\text{PolyLog}[n, a (b x^p)^q]\,dx$

   1. $\int \text{PolyLog}[n, a (b x^p)^q]\,dx$

      1: $\int \text{PolyLog}[n, a (b x^p)^q]\,dx$ when $n > 0$

- **Derivation: Integration by parts**
- **Rule: If $n > 0$, then**

$$\int \text{PolyLog}[n, a (b x^p)^q]\,dx \longrightarrow x\,\text{PolyLog}[n, a (b x^p)^q] - p\,q \int \text{PolyLog}[n-1, a (b x^p)^q]\,dx$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n,a*(b*x^p)^q] -
  p*q*Int[PolyLog[n-1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] && GtQ[n,0]
```

      2: $\int \text{PolyLog}[n, a (b x^p)^q]\,dx$ when $n < -1$

- **Derivation: Inverted integration by parts**
- **Rule: If $n < -1$, then**

$$\int \text{PolyLog}[n, a (b x^p)^q]\,dx \longrightarrow \frac{x\,\text{PolyLog}[n+1, a (b x^p)^q]}{p\,q} - \frac{1}{p\,q} \int \text{PolyLog}[n+1, a (b x^p)^q]\,dx$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  1/(p*q)*Int[PolyLog[n+1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] && LtQ[n,-1]
```

2. $\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx$

1: $\int \frac{\text{PolyLog}[n, a (b x^p)^q]}{x} \, dx$

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \text{Li}_n(z)}{\partial z} = \frac{\text{Li}_{n-1}(z)}{z}$
- **Rule:**

$$\int \frac{\text{PolyLog}[n, a (b x^p)^q]}{x} \, dx \longrightarrow \frac{\text{PolyLog}[n+1, a (b x^p)^q]}{p\, q}$$

- **Program code:**

```
Int[PolyLog[n_,c_.*(a_.+b_.*x_)^p_.]/(d_.+e_.*x_),x_Symbol] :=
  PolyLog[n+1,c*(a+b*x)^p]/(e*p) /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d,a*e]
```

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.]/x_,x_Symbol] :=
  PolyLog[n+1,a*(b*x^p)^q]/(p*q) /;
FreeQ[{a,b,n,p,q},x]
```

2. $\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx$ when $m \neq -1$

1: $\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx$ when $m \neq -1 \land n > 0$

- **Derivation: Integration by parts**
- **Rule: If** $m \neq -1 \land n > 0$**, then**

$$\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx \longrightarrow \frac{x^{m+1} \text{PolyLog}[n, a (b x^p)^q]}{m+1} - \frac{p\, q}{m+1} \int x^m \text{PolyLog}[n-1, a (b x^p)^q] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x^(m+1)*PolyLog[n,a*(b*x^p)^q]/(m+1) -
  p*q/(m+1)*Int[x^m*PolyLog[n-1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,m,p,q},x] && NeQ[m,-1] && GtQ[n,0]
```

2: $\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx$ when $m \neq -1 \land n < -1$

- **Derivation: Inverted integration by parts**
- **Rule: If** $m \neq -1 \land n < -1$**, then**

$$\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx \longrightarrow \frac{x^{m+1} \text{PolyLog}[n+1, a (b x^p)^q]}{p\, q} - \frac{m+1}{p\, q} \int x^m \text{PolyLog}[n+1, a (b x^p)^q] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x^(m+1)*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  (m+1)/(p*q)*Int[x^m*PolyLog[n+1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,m,p,q},x] && NeQ[m,-1] && LtQ[n,-1]
```

3: $\int \frac{\text{Log}[c x^m]^r \text{PolyLog}[n, a (b x^p)^q]}{x} \, dx$ when $r > 0$

- **Derivation: Integration by parts**
- **Rule: If** $r > 0$**, then**

$$\int \frac{\text{Log}[c x^m]^r \text{PolyLog}[n, a (b x^p)^q]}{x} \, dx \longrightarrow$$
$$\frac{\text{Log}[c x^m]^r \text{PolyLog}[n+1, a (b x^p)^q]}{p\, q} - \frac{m\, r}{p\, q} \int \frac{\text{Log}[c x^m]^{r-1} \text{PolyLog}[n+1, a (b x^p)^q]}{x} \, dx$$

- **Program code:**

```
Int[Log[c_.*x_^m_.]^r_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.]/x_,x_Symbol] :=
  Log[c*x^m]^r*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  m*r/(p*q)*Int[Log[c*x^m]^(r-1)*PolyLog[n+1,a*(b*x^p)^q]/x,x] /;
FreeQ[{a,b,c,m,n,q,r},x] && GtQ[r,0]
```

2. $\int u\,\text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx$

1: $\int \text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx$ when $n>0$

- **Derivation: Integration by parts**
- **Rule: If $n>0$, then**

$$\int \text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx \longrightarrow$$
$$x\,\text{PolyLog}[n,\ c\ (a+b\,x)^p]-p\int \text{PolyLog}[n-1,\ c\ (a+b\,x)^p]\,dx+a\,p\int\frac{\text{PolyLog}[n-1,\ c\ (a+b\,x)^p]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  x*PolyLog[n,c*(a+b*x)^p] -
  p*Int[PolyLog[n-1,c*(a+b*x)^p],x] +
  a*p*Int[PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x] /;
FreeQ[{a,b,c,p},x] && GtQ[n,0]
```

2. $\int (d+e\,x)^m\,\text{PolyLog}[n,\ c\ (a+b\,x)^p]\,dx$

1. $\int (d+e\,x)^m\,\text{PolyLog}[2,\ c\ (a+b\,x)]\,dx$

1: $\int \frac{\text{PolyLog}[2,\ c\ (a+b\,x)]}{d+e\,x}\,dx$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \frac{\text{PolyLog}[2,\ c\ (a+b\,x)]}{d+e\,x}\,dx \longrightarrow \frac{\text{Log}[d+e\,x]\ \text{PolyLog}[2,\ c\ (a+b\,x)]}{e}+\frac{b}{e}\int\frac{\text{Log}[d+e\,x]\ \text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[PolyLog[2,c_.*(a_.+b_.*x_)]/(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*PolyLog[2,c*(a+b*x)]/e + b/e*Int[Log[d+e*x]*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e},x]
```

2: $\int (d+e\,x)^m\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx$ when $m \neq -1$

- **Derivation: Integration by parts**
- **Rule: If $m \neq -1$, then**

$$\int (d+e\,x)^m\,\text{PolyLog}[2,\,c\,(a+b\,x)]\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,\text{PolyLog}[2,\,c\,(a+b\,x)]}{e\,(m+1)} + \frac{b}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,\text{Log}[1-a\,c-b\,c\,x]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[2,c*(a+b*x)]/(e*(m+1)) + b/(e*(m+1))*Int[(d+e*x)^(m+1)*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

x: $\int (d+e\,x)^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx$ when $n > 0 \wedge m \in \mathbb{Z}^+$

- **Derivation: Integration by parts**
- **Rule: If $n > 0 \wedge m \in \mathbb{Z}^+$, then**

$$\int (d+e\,x)^m\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,\text{PolyLog}[n,\,c\,(a+b\,x)^p]}{e\,(m+1)} - \frac{b\,p}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,\text{PolyLog}[n-1,\,c\,(a+b\,x)^p]}{a+b\,x}\,dx$$

- **Program code:**

```
(* Int[(d_.+e_.*x_)^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[n,c*(a+b*x)^p]/(e*(m+1)) -
  b*p/(e*(m+1))*Int[(d+e*x)^(m+1)*PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && GtQ[n,0] && IGtQ[m,0] *)
```

2: $\int x^m \, \text{PolyLog}[n, c (a+b x)^p] \, dx$ when $n > 0 \land m \in \mathbb{Z} \land m \neq -1$

- **Derivation: Integration by parts**
- **Rule: If** $n > 0 \land m \in \mathbb{Z} \land m \neq -1$**, then**

$$\int x^m \, \text{PolyLog}[n, c (a+b x)^p] \, dx \longrightarrow$$

$$-\frac{\left(a^{m+1} - b^{m+1} x^{m+1}\right) \text{PolyLog}[n, c (a+b x)^p]}{(m+1) \, b^{m+1}} + \frac{p}{(m+1) \, b^m} \int \text{PolyLog}[n-1, c (a+b x)^p] \, \text{ExpandIntegrand}\left[\frac{a^{m+1} - b^{m+1} x^{m+1}}{a+b x}, x\right] dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  -(a^(m+1)-b^(m+1)*x^(m+1))*PolyLog[n,c*(a+b*x)^p]/((m+1)*b^(m+1)) +
  p/((m+1)*b^m)*Int[ExpandIntegrand[PolyLog[n-1,c*(a+b*x)^p],(a^(m+1)-b^(m+1)*x^(m+1))/(a+b*x),x],x] /;
FreeQ[{a,b,c,p},x] && GtQ[n,0] && IntegerQ[m] && NeQ[m,-1]
```

3. $\int u \, \text{PolyLog}\left[n, d \left(F^{c (a+b x)}\right)^p\right] dx$

1: $\int \text{PolyLog}\left[n, d \left(F^{c (a+b x)}\right)^p\right] dx$

- **Derivation: Primitive rule**
- **Basis:** $\partial_z \text{PolyLog}[n, z] = \frac{\text{PolyLog}[n-1,z]}{z}$
- **Rule:**

$$\int \text{PolyLog}\left[n, d \left(F^{c (a+b x)}\right)^p\right] dx \longrightarrow \frac{\text{PolyLog}\left[n+1, d \left(F^{c (a+b x)}\right)^p\right]}{b \, c \, p \, \text{Log}[F]}$$

- **Program code:**

```
Int[PolyLog[n_,d_.*(F_^(c_.*(a_.+b_.*x_)))^p_.],x_Symbol] :=
  PolyLog[n+1,d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) /;
FreeQ[{F,a,b,c,d,n,p},x]
```

2: $\int (e+f\,x)^m\,\text{PolyLog}\left[n,\,d\,\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$ when $m>0$

- **Derivation: Integration by parts**
- **Basis:** $\text{PolyLog}\left[n,\,d\,\left(F^{c\,(a+b\,x)}\right)^p\right] == \partial_x \frac{\text{PolyLog}\left[n+1,d\,\left(F^{c\,(a+b\,x)}\right)^p\right]}{b\,c\,p\,\text{Log}[F]}$
- **Rule: If** $m>0$, **then**

$$\int (e+f\,x)^m\,\text{PolyLog}\left[n,\,d\,\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx \longrightarrow$$
$$\frac{(e+f\,x)^m\,\text{PolyLog}\left[n+1,\,d\,\left(F^{c\,(a+b\,x)}\right)^p\right]}{b\,c\,p\,\text{Log}[F]}-\frac{f\,m}{b\,c\,p\,\text{Log}[F]}\int (e+f\,x)^{m-1}\,\text{PolyLog}\left[n+1,\,d\,\left(F^{c\,(a+b\,x)}\right)^p\right]\,dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*PolyLog[n_,d_.*(F_^(c_.*(a_.+b_.*x_)))^p_.],x_Symbol] :=
  (e+f*x)^m*PolyLog[n+1,d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) -
  f*m/(b*c*p*Log[F])*Int[(e+f*x)^(m-1)*PolyLog[n+1,d*(F^(c*(a+b*x)))^p],x] /;
FreeQ[{F,a,b,c,d,e,f,n,p},x] && GtQ[m,0]
```

4. $\int u\,\frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx$

1: $\int \frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx$

- **Basis:** $\partial_x\,\text{PolyLog}[n+1,\,x] == \frac{\text{PolyLog}[n,x]}{x}$
- **Rule:**

$$\int \frac{\text{PolyLog}[n,\,F[x]]\,F'[x]}{F[x]}\,dx \longrightarrow \text{PolyLog}[n+1,\,F[x]]$$

- **Program code:**

```
Int[u_*PolyLog[n_,v_],x_Symbol] :=
  With[{w=DerivativeDivides[v,u*v,x]},
  w*PolyLog[n+1,v] /;
 Not[FalseQ[w]]] /;
FreeQ[n,x]
```

2: $\int \frac{\text{Log}[G[x]]\ \text{PolyLog}[n,\ F[x]]\ F'[x]}{F[x]}\, dx$

- **Derivation: Integration by parts**
- **Basis:** $\frac{\text{PolyLog}[n,x]}{x} == \partial_x \text{PolyLog}[n+1,\ x]$
- **Rule:**

$$\int \frac{\text{Log}[G[x]]\ \text{PolyLog}[n,\ F[x]]\ F'[x]}{F[x]}\, dx \longrightarrow \text{Log}[G[x]]\ \text{PolyLog}[n+1,\ F[x]] - \int \frac{G'[x]\ \text{PolyLog}[n+1,\ F[x]]}{G[x]}\, dx$$

- **Program code:**

```
Int[u_*Log[w_]*PolyLog[n_,v_],x_Symbol] :=
  With[{z=DerivativeDivides[v,u*v,x]},
  z*Log[w]*PolyLog[n+1,v] -
  Int[SimplifyIntegrand[z*D[w,x]*PolyLog[n+1,v]/w,x],x] /;
 Not[FalseQ[z]]] /;
FreeQ[n,x] && InverseFunctionFreeQ[w,x]
```